
Лабораторная работа №1 

Использование цифрового вывода 

Цель работы: ознакомиться с функциями управления выводом. 

Краткие сведения из теории 

 
Рисунок 1.1 – Принципиальная схема 

 

Листинг программы: 
void setup() 
{ 
  // настраиваем пин №13 в режим выхода, 
  // т.е. в режим источника напряжения 
  pinMode(13, OUTPUT); 
} 
  
void loop() 
{ 
  // подаём на пин 13 «высокий сигнал» (англ. «high»), т.е. 
  // выдаём 5 вольт. Через светодиод побежит ток. 
  // Это заставит его светиться 
  digitalWrite(13, HIGH); 
  
  // задерживаем (англ. «delay») микроконтроллер в этом 
  // состоянии на 100 миллисекунд 
  delay(100); 
  
  // подаём на пин 13 «низкий сигнал» (англ. «low»), т.е. 
  // выдаём 0 вольт или, точнее, приравниваем пин 13 к земле. 
  // В результате светодиод погаснет 
  digitalWrite(13, LOW); 
  
  // замираем в этом состоянии на 900 миллисекунд 
  delay(900); 
  
  // после «размораживания» loop сразу же начнёт исполняться 
  // вновь, и со стороны это будет выглядеть так, будто 
  // светодиод мигает раз в 100 мс + 900 мс = 1000 мс = 1 сек 
} 

Задание на лабораторную работу 

Светодиод должен светилтя полсекунды, а пауза между вспышками равна 

одной секунде. 

Измените код примера так, чтобы светодиод включался на три секунды по-

сле запуска устройства, а затем мигал в стандартном режиме. 

Содержание отчета 



В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Что будет, если подключить к земле анод светодиода вместо катода? 

2. Что будет, если подключить светодиод с резистором большого номинала (напри-

мер, 10 кОм)? 

3. Что будет, если подключить светодиод без резистора? 

4. Зачем нужна встроенная функция pinMode? Какие параметры она принимает? 

5. Зачем нужна встроенная функция digitalWrite? Какие параметры она принимает? 

6. С помощью какой встроенной функции можно заставить микроконтроллер ничего 

не делать? 

7. В каких единицах задается длительность паузы для этой функции? 

 

 



Лабораторная работа №2 

Использование ШИМ 

Цель работы: ознакомиться с функциями управления выводом с ШИМ. 

Краткие сведения из теории 

 
Рисунок 2.1 – Принципиальная схема 

 

Листинг программы: 
// даём разумное имя для пина №9 со светодиодом 
// (англ. Light Emitting Diode или просто «LED») 
// Так нам не нужно постоянно вспоминать куда он подключён 
#define LED_PIN 9 
  
void setup() 
{ 
  // настраиваем пин со светодиодом в режим выхода, 
  // как и раньше 
  pinMode(LED_PIN, OUTPUT); 
} 
  
void loop() 
{ 
  // выдаём неполное напряжение на светодиод 
  // (он же ШИМ-сигнал, он же PWM-сигнал). 
  // Микроконтроллер переводит число от 0 до 255 к напряжению 
  // от 0 до 5 В. Например, 85 — это 1/3 от 255, 
  // т.е. 1/3 от 5 В, т.е. 1,66 В. 
  analogWrite(LED_PIN, 85); 
  // держим такую яркость 250 миллисекунд 
  delay(250); 
  
  // выдаём 170, т.е. 2/3 от 255, или иными словами — 3,33 В. 
  // Больше напряжение — выше яркость! 
  analogWrite(LED_PIN, 170); 
  delay(250); 
  
  // все 5 В — полный накал! 
  analogWrite(LED_PIN, 255); 
  // ждём ещё немного перед тем, как начать всё заново 
  delay(250); 
} 

Задание на лабораторную работу 

Отключите питание, отключите светодиод от 9-го порта и подключите к 11-

му. Измените программу так, чтобы схема снова заработала 



Измените код программы так, чтобы в течение секунды на светодиод после-

довательно подавалось усреднённое напряжение 0, 1, 2, 3, 4, 5 В 

Возьмите еще один светодиод, резистор на 220 Ом и соберите аналогичную 

схему на этой же макетке, подключив светодиод к пину номер 3 и другому входу 

GND, измените программу так, чтобы светодиоды мигали в противофазу: первый 

выключен, второй горит максимально ярко и до противоположного состояния 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Какие из следующих идентификаторов корректны и не вызовут ошибку? 

13pin 

MOTOR_1 

контакт_светодиода 

sensor value 

leftServo 

my-var 

distance_eval2 

2. Что произойдет, если создать директиву #define HIGH LOW? 

3. Почему мы не сможем регулировать яркость светодиода, подключенного к порту 

7? 

4. Какое усреднённое напряжение мы получим на пине 6, если вызовем функцию 

analogWrite(6, 153)? 

5. Какое значение параметра value нужно передать функции analogWrite, чтобы по-

лучить усреднённое напряжение 2 В? 



Лабораторная работа №3 

Светильник с управляемой яркостью 

Цель работы: ознакомиться с принципами аналогового ввода. 

Краткие сведения из теории 

 
Рисунок 3.1 – Принципиальная схема 

 

Листинг программы: 
// даём разумные имена для пинов со светодиодом 
// и потенциометром (англ potentiometer или просто «pot») 
#define LED_PIN     9 
#define POT_PIN     A0 
  
void setup() 
{ 
  // пин со светодиодом — выход, как и раньше... 
  pinMode(LED_PIN, OUTPUT); 
  
  // ...а вот пин с потенциометром должен быть входом 
  // (англ. «input»): мы хотим считывать напряжение, 
  // выдаваемое им 
  pinMode(POT_PIN, INPUT); 
} 
  
void loop() 
{ 
  // заявляем, что далее мы будем использовать 2 переменные с 
  // именами rotation и brightness, и что хранить в них будем 
  // целые числа (англ. «integer», сокращённо просто «int») 
  int rotation, brightness;   
  
  // считываем в rotation напряжение с потенциометра: 
  // микроконтроллер выдаст число от 0 до 1023 
  // пропорциональное углу поворота ручки 
  rotation = analogRead(POT_PIN); 
  
  // в brightness записываем полученное ранее значение rotation 
  // делённое на 4. Поскольку в переменных мы пожелали хранить 
  // целые значения, дробная часть от деления будет отброшена. 
  // В итоге мы получим целое число от 0 до 255 
  brightness = rotation / 4; 
  
  // выдаём результат на светодиод 
  analogWrite(LED_PIN, brightness); 
} 

Задание на лабораторную работу 



Отключите питание платы, подключите к порту 5 еще один светодиод. Из-

мените код таким образом, чтобы второй светодиод светился на 1/8 от яркости пер-

вого 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Можем ли мы при сборке схемы подключить светодиод и потенциометр напря-

мую к разным входам GND микроконтроллера? 

2. В какую сторону нужно крутить переменный резистор для увеличения яркости 

светодиода? 

3. Что будет, если стереть из программы строчку pinMode(LED_PIN, OUTPUT)? 

строчку pinMode(POT_PIN, INPUT)? 

4. Зачем мы делим значение, полученное с аналогового входа перед тем, как задать 

яркость светодиода? что будет, если этого не сделать? 



Лабораторная работа №4 

Управление вводом/выводом 

Цель работы: ознакомиться с принципами ввода/вывода. 

Краткие сведения из теории 

 
Рисунок 4.1 – Принципиальная схема 

 

Листинг программы: 
// даём имена для пинов с пьезопищалкой (англ. buzzer) и фото- 
// резистором (англ. Light Dependent Resistor или просто LDR) 
#define BUZZER_PIN  3 
#define LDR_PIN     A0 
  
void setup() 
{ 
  // пин с пьезопищалкой — выход... 
  pinMode(BUZZER_PIN, OUTPUT); 
  
  // ...а все остальные пины являются входами изначально, 
  // всякий раз при подаче питания или сбросе микроконтроллера. 
  // Поэтому, на самом деле, нам совершенно необязательно 
  // настраивать LDR_PIN в режим входа: он и так им является 
} 
  
void loop() 
{ 
  int val, frequency; 
  
  // считываем уровень освещённости так же, как для 
  // потенциометра: в виде значения от 0 до 1023. 
  val = analogRead(LDR_PIN); 
  
  // рассчитываем частоту звучания пищалки в герцах (ноту), 
  // используя функцию проекции (англ. map). Она отображает 
  // значение из одного диапазона на другой, строя пропорцию. 
  // В нашем случае [0; 1023] -> [3500; 4500]. Так мы получим 
  // частоту от 3,5 до 4,5 кГц. 
  frequency = map(val, 0, 1023, 3500, 4500); 
  
  // заставляем пин с пищалкой «вибрировать», т.е. звучать 
  // (англ. tone) на заданной частоте 20 миллисекунд. При 
  // cледующих проходах loop, tone будет вызван снова и снова, 
  // и на деле мы услышим непрерывный звук тональностью, которая 
  // зависит от количества света, попадающего на фоторезистор 
  tone(BUZZER_PIN, frequency, 20); 
} 

Задание на лабораторную работу 



Уберите из программы чтение датчика освещенности и пропищите азбукой 

Морзе позывной SOS: три точки, три тире, три точки 

Измените код программы так, чтобы с падением освещенности звук стано-

вился ниже (например, падал от 5 кГц до 2,5 кГц) 

Измените код программы так, чтобы звук терменвокса раздавался не непре-

рывно, а 10 раз в секунду с различимыми паузами 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Каким сопротивлением должен обладать фоторезистор, чтобы на аналоговый 

вход было подано напряжение 1 В? 

2. Можем ли мы регулировать яркость светодиода, подключенного к 11-му порту, во 

время звучания пьезопищалки? 

3. Что изменится в работе терменвокса, если заменить резистор на 10 кОм резисто-

ром на 100 кОм? Попробуйте ответить без эксперимента. Затем отключите пита-

ние, замените резистор и проверьте. 

4. Каков будет результат вызова map(30,0,90,90,-90)? 

5. Как будет работать вызов tone без указания длительности звучания? 

6. Можно ли устроить полифоническое звучание с помощью функции tone? 



Лабораторная работа №5 

Управление выводом по событию на вводе 

Цель работы: ознакомиться с принципами ввода/вывода. 

Краткие сведения из теории 

 
Рисунок 5.1 – Принципиальная схема 

 

Листинг программы: 
#define LED_PIN  13 
#define LDR_PIN  A0 
#define POT_PIN  A1 
void setup() 
{ 
  pinMode(LED_PIN, OUTPUT); 
} 
  
void loop() 
{ 
  // считываем уровень освещённости. Кстати, объявлять 
  // переменную и присваивать ей значение можно разом 
  int lightness = analogRead(LDR_PIN); 
  // считываем значение с потенциометра, которым мы регулируем 
  // пороговое значение между условными темнотой и светом 
  int threshold = analogRead(POT_PIN); 
  // объявляем логическую переменную и назначаем ей значение 
  // «темно ли сейчас». Логические переменные, в отличие от 
  // целочисленных, могут содержать лишь одно из двух значений: 
  // истину (англ. true) или ложь (англ. false). Такие значения 
  // ещё называют булевыми (англ. boolean). 
  boolean tooDark = (lightness < threshold); 
  // используем ветвление программы: процессор исполнит один из 
  // двух блоков кода в зависимости от исполнения условия. 
  // Если (англ. «if») слишком темно... 
  if (tooDark) { 
    // ...включаем освещение 
    digitalWrite(LED_PIN, HIGH); 
  } else { 
    // ...иначе свет не нужен — выключаем его 
    digitalWrite(LED_PIN, LOW); 
  } 
} 



Задание на лабораторную работу 

Перепишите программу без использования переменной tooDark с сохранени-

ем функционала устройства. 

Добавьте в схему еще один светодиод. Дополните программу так, чтобы при 

падении освещенности ниже порогового значения включался один светодиод, а при 

падении освещенности ниже половины от порогового значения включались оба све-

тодиода. 

Измените схему и программу так, чтобы светодиоды включались по преж-

нему принципу, но светились тем сильнее, чем меньше света падает на фоторези-

стор. 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Если мы установим фоторезистор между аналоговым входом и землей, наше 

устройство будет работать наоборот: светодиод будет включаться при увеличении 

количества света. Почему? 

2. Какой результат работы устройства мы получим, если свет от светодиода будет 

падать на фоторезистор? 

3. Если мы все же установили фоторезистор так, как сказано в предыдущем вопросе, 

как нам нужно изменить программу, чтобы устройство работало верно? 

4. Допустим, у нас есть код if (условие) {действие;}. В каких случаях будет выпол-

нено действие? 

5. При каких значениях y выражение x + y > 0 будет истинным, если x > 0? 

6. Обязательно ли указывать, какие инструкции выполнять, если условие в операто-

ре if ложно? 

7. Чем отличается оператор == от оператора =? 

8. Если мы используем конструкцию if (условие) действие1; else действие2;, может 

ли быть ситуация, когда ни одно из действий не выполнится? Почему? 



Лабораторная работа №6 

Управление таймером с выводом на семисегментный индикатор 

Цель работы: ознакомиться с принципами вывода. 

Краткие сведения из теории 

 
Рисунок 6.1 – Принципиальная схема 

 

Листинг программы: 
#define FIRST_SEGMENT_PIN   2 
#define SEGMENT_COUNT       7 
// префикс «0b» означает, что целое число за ним записано в 
// в двоичном коде. Единицами мы обозначим номера сегментов 
// индикатора, которые должны быть включены для отображения 
// арабской цифры. Всего цифр 10, поэтому в массиве 10 чисел. 
// Нам достаточно всего байта (англ. byte, 8 бит) для хранения 
// комбинации сегментов для каждой из цифр. 
byte numberSegments[10] = { 
  0b00111111, 0b00001010, 0b01011101, 0b01011110, 0b01101010, 
  0b01110110, 0b01110111, 0b00011010, 0b01111111, 0b01111110, 
}; 
void setup() 
{ 
  for (int i = 0; i < SEGMENT_COUNT; ++i) 
    pinMode(i + FIRST_SEGMENT_PIN, OUTPUT); 
} 
void loop() 
{ 
  // определяем число, которое собираемся отображать. Пусть им 
  // будет номер текущей секунды, зацикленный на десятке 
  int number = (millis() / 1000) % 10; 
  // получаем код, в котором зашифрована арабская цифра 
  int mask = numberSegments[number]; 
  // для каждого из 7 сегментов индикатора... 
  for (int i = 0; i < SEGMENT_COUNT; ++i) { 
    // ...определяем: должен ли он быть включён. Для этого 
    // считываем бит (англ. read bit), соответствующий текущему 
    // сегменту «i». Истина — он установлен (1), ложь — нет (0) 
    boolean enableSegment = bitRead(mask, i); 
    // включаем/выключаем сегмент на основе полученного значения 
    digitalWrite(i + FIRST_SEGMENT_PIN, enableSegment); 
  } 
} 



Задание на лабораторную работу 

Измените код, чтобы индикатор отсчитывал десятые секунды. 

Поменяйте программу так, чтобы вместо символа «0» отображался символ 

«А». 

Дополните схему и программу таким образом, чтобы сегмент-точка вклю-

чался при прохождении четных чисел и выключался на нечетных 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. К которой ножке нашего семисегментного индикатора нужно подключать зем-

лю? 

2. Как мы храним закодированные символы цифр? 

3. Каким образом мы выводим символ на индикатор? 



Лабораторная работа №7 

Управление аналоговым выводом с помощью цифрового ввода 

Цель работы: ознакомиться с принципами ввода и вывода. 

Краткие сведения из теории 

 
Рисунок 7.1 – Принципиальная схема 

 

Листинг программы: 
#define PLUS_BUTTON_PIN     2 
#define MINUS_BUTTON_PIN    3 
#define LED_PIN             9 
int brightness = 100; 
boolean plusUp = true; 
boolean minusUp = true; 
void setup() 
{ 
  pinMode(LED_PIN, OUTPUT); 
  pinMode(PLUS_BUTTON_PIN, INPUT_PULLUP); 
  pinMode(MINUS_BUTTON_PIN, INPUT_PULLUP); 
} 
void loop() 
{ 
  analogWrite(LED_PIN, brightness); 
  // реагируем на нажатия с помощью функции, написанной нами 
  plusUp = handleClick(PLUS_BUTTON_PIN, plusUp, +35); 
  minusUp = handleClick(MINUS_BUTTON_PIN, minusUp, -35); 
} 
// Собственная функция с 3 параметрами: номером пина с кнопкой 
// (buttonPin), состоянием до проверки (wasUp) и градацией 
// яркости при клике на кнопку (delta). Функция возвращает 
// (англ. return) обратно новое, текущее состояние кнопки 
boolean handleClick(int buttonPin, boolean wasUp, int delta) 
{ 
  boolean isUp = digitalRead(buttonPin); 
  if (wasUp && !isUp) { 
    delay(10); 
    isUp = digitalRead(buttonPin); 
    // если был клик, меняем яркость в пределах от 0 до 255 
    if (!isUp)     
      brightness = constrain(brightness + delta, 0, 255); 
  } 
  return isUp; // возвращаем значение обратно, в вызывающий код 
} 

Задание на лабораторную работу 



Доработайте код таким образом, чтобы шаг изменения яркости настраивался 

в одном месте. 

Создайте еще одну функцию и переделайте код так, чтобы одна функция от-

вечала за отслеживание нажатий, а другая — за вычисление яркости светодиода и 

возвращала его в analogWrite. 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Что необходимо для определения собственной функции? 

2. Что означает ключевое слово void? 

3. Как ведет себя программа при упоминании одной переменной с разных сторон от 

оператора присваивания =? 

Практическое занятие №1 

Цифровые выводы 

 

Выводы платформы Arduino могут работать как входы или как выходы. Дан-

ный документ объясняет функционирование выводов в этих режимах. Также необ-

ходимо обратить внимание на то, что большинство аналоговых входов Arduino 

(Atmega) могут конфигурироваться и работать так же как и цифровые порты вво-

да/вывода. 

Свойства порта вводы/вывода (pin), сконфигурированного как порт ввода 

Выводы Arduino (Atmega) стандартно настроены как порты ввода, таким об-

разом, не требуется явной декларации в функции pinMode(). Сконфигурированные 

порты ввода находятся в высокоимпедансном состоянии. Это означает то, что порт 

ввода дает слишком малую нагрузки на схему, в которую он включен. Эквивален-

том внутреннему сопротивлению будет резистор 100 МОм подключенный к выводу 

микросхемы. Таким образом, для перевода порта ввода из одного состояния в дру-

гое требуется маленькое значение тока. Это позволяет применять выводы микро-

схемы для подключения емкостного датчика касания, фотодиода, аналогового дат-

чика со схемой, похожей на RC-цепь. 

С другой стороны, если к данному выводу ничего не подключено, то значе-

ния на нем будут принимать случайные величины, наводимые электрическими по-

мехами или емкостной взаимосвязью с соседним выводом. 

Подтягивающие (нагрузочные) резисторы 

Если на порт ввода не поступает сигнал, то в данном случае рекомендуется 

задать порту известное состояние. Это делается добавлением подтягивающих рези-

сторов 10 кОм, подключающих вход либо к +5 В (подтягивающие к питанию рези-

сторы), либо к земле (подтягивающие к земле резисторы). 

Микроконтроллер Atmega имеет программируемые встроенные подтягива-

ющие к питанию резисторы 20 кОм. Программирование данных резисторов осу-

ществляется следующим образом. 

pinMode(pin, INPUT);           // назначить выводу порт ввода 

digitalWrite(pin, HIGH);       // включить подтягивающий резистор 



Подтягивающий резистор пропускает ток достаточный для того, чтобы слег-

ка светился светодиод подключенный к выводу, работающему как порт ввода. Так-

же легкое свечение светодиодов означает то, что при программировании вывод не 

был настроен как порт вывода в функции pinMode(). 

Подтягивающие резисторы управляются теми же регистрами (внутренние 

адреса памяти микроконтроллера), что управляют состояниями вывода: HIGH или 

LOW. Следовательно, если вывод работает как порт ввода со значением HIGH, это 

означает включение подтягивающего к питанию резистора, то конфигурация функ-

цией pinMode() порта вывода на данном выводе микросхемы передаст значение 

HIGH. Данная процедура работает и в обратном направлении, т.е. если вывод имеет 

значение HIGH, то конфигурация вывода микросхемы как порта ввода функцией 

pinMode() включит подтягивающий к питанию резистор. 

Примечание: Затруднительно использовать вывод микросхемы 13 в качестве 

порта ввода из-за подключенных к нему светодиода и резистора. При подключении 

подтягивающего к питанию резистора 20 кОм на вводе будет 1.7 В вместо 5 В, т.к. 

происходит падение напряжения на светодиоде и включенном последовательно ре-

зисторе. При необходимости использовать вывод микросхемы 13 как цифровой порт 

ввода требуется подключить между выводом и землей внешний подтягивающий ре-

зистор. 

Свойства порта ввода/вывода, сконфигурированного как порт вывода 

Выводы, сконфигурированные как порты вывода, находятся в низкоимпе-

дансном состоянии. Данные выводы могут пропускать через себя достаточно боль-

шой ток. Выводы микросхемы Atmega могут быть источником (положительный) 

или приемником (отрицательный) тока до 40 мА для других устройств. Такого зна-

чения тока достаточно чтобы подключить светодиод (обязателен последовательно 

включенный резистор), датчики, но недостаточно для большинства реле, соленоидов 

и двигателей. 

Короткие замыкания выводов Arduino или попытки подключить энергоемкие 

устройства могут повредить выходные транзисторы вывода или весь микроконтрол-

лер Atmega. В большинстве случаев данные действия приведут к отключению выво-

да на микроконтроллере, но остальная часть схемы будет работать согласно про-



грамме. Рекомендуется к выходам платформы подключать устройства через рези-

сторы 470 Ом или 1 кОм, если устройству не требуется больший ток для работы.  



Практическое занятие №2 

Аналоговые входы 

 

Описание портов, работающих как аналоговые входы, платформы Arduino 

(Atmega8, Atmega168, Atmega328, или Atmega1280) 

Аналого-цифровой преобразователь 

Микроконтроллеры Atmega, используемые в Arduino, содержат шестика-

нальный аналого-цифровой преобразователь (АЦП). Разрешение преобразователя 

составляет 10 бит, что позволяет на выходе получать значения от 0 до 1023. Основ-

ным применением аналоговых входов большинства платформ Arduino является чте-

ние аналоговых датчиком, но в тоже время они имеют функциональность вво-

дов/выводов широкого применения (GPIO) (то же, что и цифровые порты вво-

да/вывода 0 - 13). 

Таким образом, при необходимости применения дополнительных портов 

ввода/вывода имеется возможность сконфигурировать неиспользуемые аналоговые 

входы. 

Цоколевка 

Выводы Arduino, соответствующие аналоговым входам, имеют номера от 14 

до 19. Это относится только к выводам Arduino, а не к физическим номерам выводов 

микроконтроллера Atmega. Аналоговые входы могут использоваться как цифровые 

выводы портов ввода/вывода. Например, код программы для установки вывода 0 

аналогового входа на порт вывода со значением HIGH: 

pinMode(14, OUTPUT); 

digitalWrite(14, HIGH); 

Подтягивающие резисторы 

Выводы аналоговые входов имеют подтягивающие резисторы работающие 

как на цифровых выводах. Включение резисторов производится командой 

digitalWrite(14, HIGH); // включить резистор на выводе аналогового входа 0 

пока вывод работает как порт ввода. 

Подключение резистора повлияет на величину сообщаемую функцией 

analogRead() при использовании некоторых датчиков. Большинство пользователей 



использует подтягивающий резистор при применении вывода аналогового входа в 

его цифровом режиме. 

Подробности и предостережения 

Для вывода, работавшего ранее как цифровой порт вывода, команда 

analogRead будет работать некорректно. В этом случае рекомендуется сконфигури-

ровать его как аналоговый вход. Аналогично, если вывод работал как цифровой 

порт вывода со значением HIGH, то обратная установка на ввод подключит подтя-

гивающий резистор.  

Руководство на микроконтроллер Atmega не рекомендует производить быст-

рое переключение между аналоговыми входами для их чтения. Это может вызвать 

наложение сигналов и внести искажения в аналоговую систему. Однако после рабо-

ты аналогового входа в цифровом режиме может потребоваться настроить паузу 

между чтением функцией analogRead() других входов. 



Практическое занятие №3 

Широтно-импульсная модуляция 

 

Широтно-Импульсная модуляция, сокращенно ШИМ (англ. PWM) 

 

Пример использования аналогового выхода (ШИМ) для управления светоди-

одом доступен из меню File->Sketchbook->Examples->Analog программы Arduino. 

Широтно-Импульсная модуляция, или ШИМ, это операция получения изме-

няющегося аналогового значения посредством цифровых устройств. Устройства ис-

пользуются для получения прямоугольных импульсов - сигнала, который постоянно 

переключается между максимальным и минимальным значениями.   Данный сигнал 

моделирует напряжение между максимальным значением (5 В) и минимальным (0 

В), изменяя при этом длительность времени включения 5 В относительно включе-

ния 0 В. Длительность включения максимального значения называется шириной 

импульса. Для получения различных аналоговых величин изменяется ширина им-

пульса. При достаточно быстрой смене периодов включения-выключения можно 

подавать постоянный сигнал между 0 и 5 В на светодиод, тем самым управляя ярко-

стью его свечения. 

На графике зеленые линии отмечают постоянные временные периоды. Дли-

тельность периода обратно пропорциональна частоте ШИМ. Т.е. если частота ШИМ 

составляет 500 Гц, то зеленые линии будут отмечать интервалы длительностью в 2 

миллисекунды каждый. Вызов функции analogWrite() с масштабом 0 – 255 означает, 

что значение analogWrite(255) будет соответствовать 100% рабочему циклу (посто-

янное включение 5 В), а значение analogWrite(127) – 50% рабочему циклу.  



 
Рисунок 3.1 – Широтно-импульсная модуляция ШИМ PWM. 

 

Для примера можно взять платформу и начать трясти ее взад и вперед. Для 

наших глаз данное движение превращает в светящиеся линии мигание светодиода. 

Наращивание или уменьшение ширины импульса на светодиоде будет увеличивать 

или уменьшать светящиеся линии светодиода. 



Практическое занятие №4 

Память в Arduino 

 

В микроконтроллере ATmega168, используемом на платформах Arduino, су-

ществует три вида памяти: 

Флеш-память: используется для хранения скетчей. 

ОЗУ (Статическая оперативная память с произвольным доступом): исполь-

зуется для хранения и работы переменных.  

EEPROM (энергонезависимая память): используется для хранения постоян-

ной информации. 

Флеш-память и EEPROM являются энергонезависимыми видами памяти 

(данные сохраняются при отключении питания). ОЗУ является энергозависимой па-

мятью. 

Микроконтроллер ATmega168 имеет: 

16 Кб флеш-памяти (2 Кб используется для хранения загрузчика) 

1024 байта ОЗУ 

512 байт EEPROM 

Необходимо обратить внимание на малый объем ОЗУ, т.к. большое число 

строк в скетче может полностью ее израсходовать. Например, следующая объявле-

ние: 

char message[] = "I support the Cape Wind project."; 

занимает 32 байта из общего объема ОЗУ (каждый знак занимает один байт). 

При наличии большого объема текста или таблиц для вывода на дисплей возможно 

полностью использовать допустимые 1024 байта ОЗУ. 

При отсутствии свободного места в ОЗУ могут произойти сбои программы, 

например, она может записаться, но не работать. Для определения данного состоя-

ния требуется превратить в комментарии или укоротить строки скетча (без измене-

ния кода). Если после этого программа работает корректно, то на ее выполнение был 

затрачен весь объем ОЗУ. Существует несколько путей решения данной проблемы: 

При работе скетча с программой на компьютере можно перебросить часть 

данных или расчетов на компьютер для снижения нагрузки на Arduino. 



При наличии таблиц поиска или других больших массивов можно использо-

вать минимальный тип данных для хранения значений. Например, тип данных зани-

мает два байта, а byte - только один (но может хранить небольшой диапазон значе-

ний). 

Неизменяемые строки и данные во время работы скетча можно хранить во 

флеш-памяти. Для этого необходимо использовать ключ PROGMEM. 

Для использования EEPROM обратитесь к библиотеке EEPROM. 



Практическое занятие №5 

Использование прерываний в Arduino 

 

Часто при работе с проектами на микроконтроллерах требуется запускать 

фоновую функцию через равные промежутки времени. Это часто реализуется уста-

новкой аппаратного таймера для выработки прерывания. Это прерывание запускает 

программу обработки прерываний (Interrupt Service Routine, ISR) для управления 

периодическим прерыванием. В настоящей статье я описываю установку 8-битного 

таймера 2 для выработки прерываний на микроконтроллере ATMega168 Arduino. Я 

пройдусь по этапам, требуемым для установки программы обработки прерываний и 

внутри нее самой. 

Arduino подразумевает процессор ATMega168. Этот микроконтроллер имеет 

несколько систем ввода-вывода, которые доступны каждому пользователю Arduino, 

поскольку библиотека Arduino облегчает их использование. К примеру, цифровой 

ввод-вывод, ШИМ, аналого-цифровые входы и последовательный порт. ATMega168 

также имеет три внутренних аппаратных таймера. Хотя библиотека Arduino позво-

ляет использовать некоторые свойства таймеров, нельзя напрямую использовать 

таймер для выработки периодических прерываний. 

Как подсказывает название, прерывания – это сигналы, прерывающие нор-

мальное течение программы. Прерывания обычно используются для аппаратных 

устройств, требующих немедленной реакции на появление событий. Например, си-

стема последовательного порта или UART (универсальный асинхронный приемопе-

редатчик) микроконтроллера должен быть обслужен при получении нового символа. 

Если этого не сделать быстро, новый символ может быть потерян. 

При поступлении нового символа UART генерирует прерывание. Микро-

контроллер останавливает выполнение основной программы (вашего приложения) и 

перескакивает на программу обработки прерываний (ISR), предназначенную для 

данного прерывания. В данном случае это прерывание по полученному символу. 

Эта ISR захватывает новый символ из UART, помещает в буфер, затем очищает 

прерывание и выполняет возврат. Когда ISR выполняет возврат, микроконтроллер 

возвращается в основную программу и продолжает её с точки вызова. Все это про-



исходит в фоновом режиме и не влияет напрямую на основной код вашего прило-

жения. 

Если у вас запускается много прерываний или прерывания генерирует быст-

родействующий таймер, ваша основная программа будет выполняться медленнее, 

так как микроконтроллер распределяет свое машинное время между основной про-

граммой и всеми функциями обработки прерываний. 

Вы можете задуматься, почему бы не просто проверять новый символ время 

от времени, вместо использования такого сложного процесса прерывания. Давайте 

вычислим пример, чтобы увидеть, насколько важны процессы прерывания. Скажем, 

у вас есть последовательный порт со скоростью передачи данных 9600 бод. Это 

означает, что каждый бит символа посылается с частотой 9600 Гц или около 10 кГц. 

На каждый бит уходит 100 мкс. Около 10 бит требуется, чтобы послать один сим-

вол, так что мы получаем один полный символ каждую миллисекунду или около то-

го. Если наш UART буферизован, мы должны извлечь последний символ до завер-

шения приема следующего, это дает нам на всю работу 1 мс. Если наш UART не 

буферизован, мы должны избавиться от символа за 1 бит или 1 мкс. Рассмотрим для 

начала буферизованный пример. 

Мы должны проверять получение байта быстрее, чем каждую миллисекунду, 

чтобы предотвратить потерю данных. Применительно к Arduino это означает, что 

наша функция цикла должна обращаться для чтения статуса UART и возможно, 

байта данных, 1000 раз в секунду. Это легко выполнимо, но сильно усложнит код, 

который вам нужно написать. До тех пор, пока ваша функция цикла не требует 

больше 1 мс до завершения, вам это может сойти с рук. Но представьте, что вам 

нужно обслуживать несколько устройств ввода-вывода, или что необходимо рабо-

тать на гораздо большей скорости передачи. Видите, какие неприятности это вскоре 

может принести. 

С прерываниями вам не нужно отслеживать поступление символа. Аппара-

тура подает сигнал с помощью прерывания, и процессор быстро вызовет ISR, чтобы 

вовремя захватить символ. Вместо выделения огромной доли процессорного време-

ни на проверку статуса UART, вы никогда не должны проверять статус, вы просто 

устанавливаете аппаратное прерывание и выполняете необходимые действия в ISR. 



Ваша главная программа напрямую не затрагивается, и от аппаратного устройства 

не требуется особых возможностей. 

Прерывание по таймеру 

В настоящей статье я сосредоточусь на использовании программного тайме-

ра 2 для периодических прерываний. Исходная идея состояла в использовании этого 

таймера для генерации частоты биений в звуковых проектах Arduino. Чтобы выво-

дить тон или частоту нам нужно переключать порт ввода-вывода на согласованной 

частоте. Это можно делать с использованием циклов задержки. Это просто, но озна-

чает, что наш процессор будет занят, ничего не выполняя, но ожидая точного вре-

мени переключения вывода. С использованием прерывания по таймеру мы можем 

заняться другими делами, а вывод пусть переключает ISR, когда таймер подаст сиг-

нал, что время пришло. 

Нам нужно только установить таймер, чтобы подавал сигнал с прерыванием 

в нужное время. Вместо прокрутки бесполезного цикла для задержки по времени, 

наша главная программа может делать что-то другое, например, контролировать 

датчик движения или управлять электроприводом. Что бы ни требовалось нашему 

проекту, больше нам не нужно процессорное время для получения задержек. 



Практическое занятие №6 

Переменные 

 

Переменная – это место хранения данных. Она имеет имя, значение и тип. 

Например, данное объявление (называется декларацией): 

int pin = 13; 

создает переменную с именем pin, значением 13 и типом int.   Затем в про-

грамме имеется возможность обратиться к данной переменной через имя с целью 

работы с ее значением. Например, в утверждении: 

pinMode(pin, OUTPUT); 

имеется значение вывода (13), которое будет передаваться в функцию 

pinMode(). В данном случае нет необходимости использовать переменную. Утвер-

ждение может работать и в таком виде: 

pinMode(13, OUTPUT); 

Преимущество переменной заключается в том, что необходимо определить 

значение вывода однажды и потом использовать его многократно. В последствии 

при изменении вывода 13 на 12 достаточно будет поменять только одну строку в 

программном коде. Также можно использовать специальные имена для подчеркива-

ния значения переменной (напр., программа, управляющая светодиодом RGB, мо-

жет содержать переменные redPin, greenPin и bluePin). 

Переменные имеют другие преимущества перед такими значениями как чис-

ло. Имеется возможность изменить значение переменной, используя присвоение. 

Например:  

pin = 12; 

изменит значение переменной на число 12. В данном примере не определя-

ется тип переменной, т.к. он не меняется операцией присвоения. Имя переменной 

постоянно связано с типом, меняется только значение. [1] Перед присвоением зна-

чения необходимо декларировать переменную. Присвоение значения переменной 

без ее декларации вызовет следующее сообщение: error: pin was not declared in this 

scope". 

 



При присвоении одной переменной другой происходит копирование значе-

ния первой переменной во вторую. Изменение значение одной переменной не влия-

ет на другую. Например, после записи: 

int pin = 13; 

int pin2 = pin; 

pin = 12; 

только pin имеет значение 12, а pin2 еще равен 13. 

Что означает слово "scope" в сообщение об ошибке, приведенной выше? Оно 

относится к части программы, в которой переменная может использоваться - обла-

сте видимости. Область видимости определяется местом ее декларации. Например, 

имеется возможность использовать переменную во всей программе, если задеклари-

ровать ее в начале программного кода. Такие переменные называются глобальными. 

Например: 

int pin = 13; 

void setup() 

{ 

  pinMode(pin, OUTPUT); 

} 

void loop() 

{ 

  digitalWrite(pin, HIGH); 

} 

Из примера видно, что pin используется в обеих функциях setup() и loop().   

Обе функции ссылаются на одну переменную, таким образом, изменение ее значе-

ния в одной функции повлияет на значение в другой:  

int pin = 13; 

void setup() 

{ 

  pin = 12; 

  pinMode(pin, OUTPUT); 

} 



void loop() 

{ 

  digitalWrite(pin, HIGH); 

} 

Функции digitalWrite(), вызываемой из loop(), будет передано значение 12, 

т.к. оно было присвоено переменной в функции setup().  

Если переменная используется только один раз в функции, то ее деклариру-

ют в данной части программного кода, ограниченной скобками функции. Например: 

void setup() 

{ 

  int pin = 13; 

  pinMode(pin, OUTPUT); 

  digitalWrite(pin, HIGH); 

} 

В данном примере переменная может использоваться только внутри функ-

ции setup(). При написании данного кода: 

void loop() 

{ 

  digitalWrite(pin, LOW); // wrong: pin is not in scope here. 

} 

будет выведено сообщение: «error: 'pin' was not declared in this scope». Данное 

сообщение будет выводиться, даже если вы задекларировали переменную где-то в 

программе, но пытаетесь ее использовать вне области видимости. 

Почему не сделать все переменные глобальными? Если неизвестно где будет 

еще использоваться переменная, то почему ее надо ограничивать одной функцией? 

Когда переменная ограничена легче найти источник ее изменения. Если переменная 

глобальная, то ее значение может быть изменено в любом месте программного кода, 

что означает необходимость проследить ее изменение по всей программе. Напри-

мер, когда переменная имеет некорректное значение, то гораздо легче найти причи-

ну если область видимости ограничена. 



Практическое занятие №6 

Функции 

 

Разбиение на сегменты кода функциями позволяет создавать части кода, ко-

торые выполняют определенные задания. После выполнения происходит возврат в 

место, откуда была вызвана функция. Причиной создания функции является необ-

ходимость выполнять одинаковое действие несколько раз. 

Для программистов, работающих с BASIC, функции в Arduino позволяют 

использовать подпрограммы (GOSUB в BASIC). 

Разделения кода на функции имеет ряд преимуществ: 

Функции позволяют организовать программу. Очень часто помогают заранее 

составить концепцию программы. 

Функции кодируют одно действие в одном месте программы. Далее необхо-

димо только отладить код функции. 

Функции сокращают шансы на появление ошибки при необходимости изме-

нения кода. 

Функции сокращают текст скетчей и делают его компактным, т.к. некоторые 

секции используются много раз. 

Функции облегчают использование кода в других программах делая его мо-

дульным. В этом случае функции обладают еще одним небольшим преимуществом, 

делая код программы легким для чтения. 

Существуют две обязательные функции в скетчах Arduino setup() и loop(). 

Другие функции должны создаваться за скобками этих функций. В следующем при-

мере будет создана простая функция умножения двух чисел. 

Пример 



 
Рисунок 7.1 – Синтаксис функции Arduino 

 

Для вызова функции умножения ей передаются параметры данных: 

void loop(){ 

   int i = 2; 

   int j = 3; 

   int k; 

   k = myMultiplyFunction(i, j); // k содержит 6 

} 

Созданную функцию необходимо задекларировать вне скобок любой другой 

функции, таким образом "myMultiplyFunction()" может стоять выше или ниже функ-

ции "loop()". 

Весь скетч будет выглядеть следующим образом: 

void setup(){ 

  Serial.begin(9600); 

} 

void loop(){ 

  int i = 2; 

  int j = 3; 

  int k; 

  k = myMultiplyFunction(i, j); // k содержит 6 

  Serial.println(k); 



  delay(500); 

} 

int myMultiplyFunction(int x, int y){ 

  int result; 

  result = x * y; 

  return result; 

} 

Следующая функция будет считывать данные с датчика функцией 

analogRead() и затем рассчитывать среднее арифметическое. Затем созданная функ-

ция будет масштабировать данные по 8 битам (0-255) и инвертировать их. // датчик 

подключен к выводу 0 

int ReadSens_and_Condition(){ 

  int i; 

  int sval; 

  for (i = 0; i < 5; i++){ 

    sval = sval + analogRead(0);    // сенсор на аналоговом входе 0 

  } 

  sval = sval / 5;    // среднее 

  sval = sval / 4;    // масштабирование по 8 битам (0 - 255) 

  sval = 255 - sval;  // инвертирование выходного значения 

  return sval; 

} 

Вызов функции осуществляется присвоением ее переменной. 

int sens; 

sens = ReadSens_and_Condition(); 

Лабораторная работа №1 

Использование цифрового вывода 

Цель работы: ознакомиться с функциями управления выводом. 

Краткие сведения из теории 

 
Рисунок 1.1 – Принципиальная схема 

 

Листинг программы: 
void setup() 
{ 
  // настраиваем пин №13 в режим выхода, 
  // т.е. в режим источника напряжения 
  pinMode(13, OUTPUT); 
} 
  
void loop() 
{ 
  // подаём на пин 13 «высокий сигнал» (англ. «high»), т.е. 
  // выдаём 5 вольт. Через светодиод побежит ток. 
  // Это заставит его светиться 
  digitalWrite(13, HIGH); 
  
  // задерживаем (англ. «delay») микроконтроллер в этом 
  // состоянии на 100 миллисекунд 
  delay(100); 
  
  // подаём на пин 13 «низкий сигнал» (англ. «low»), т.е. 
  // выдаём 0 вольт или, точнее, приравниваем пин 13 к земле. 
  // В результате светодиод погаснет 
  digitalWrite(13, LOW); 
  
  // замираем в этом состоянии на 900 миллисекунд 
  delay(900); 
  
  // после «размораживания» loop сразу же начнёт исполняться 
  // вновь, и со стороны это будет выглядеть так, будто 
  // светодиод мигает раз в 100 мс + 900 мс = 1000 мс = 1 сек 
} 

Задание на лабораторную работу 

Светодиод должен светилтя полсекунды, а пауза между вспышками равна 

одной секунде. 

Измените код примера так, чтобы светодиод включался на три секунды по-

сле запуска устройства, а затем мигал в стандартном режиме. 

Содержание отчета 



В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Что будет, если подключить к земле анод светодиода вместо катода? 

2. Что будет, если подключить светодиод с резистором большого номинала (напри-

мер, 10 кОм)? 

3. Что будет, если подключить светодиод без резистора? 

4. Зачем нужна встроенная функция pinMode? Какие параметры она принимает? 

5. Зачем нужна встроенная функция digitalWrite? Какие параметры она принимает? 

6. С помощью какой встроенной функции можно заставить микроконтроллер ничего 

не делать? 

7. В каких единицах задается длительность паузы для этой функции? 

 

 



Лабораторная работа №2 

Использование ШИМ 

Цель работы: ознакомиться с функциями управления выводом с ШИМ. 

Краткие сведения из теории 

 
Рисунок 2.1 – Принципиальная схема 

 

Листинг программы: 
// даём разумное имя для пина №9 со светодиодом 
// (англ. Light Emitting Diode или просто «LED») 
// Так нам не нужно постоянно вспоминать куда он подключён 
#define LED_PIN 9 
  
void setup() 
{ 
  // настраиваем пин со светодиодом в режим выхода, 
  // как и раньше 
  pinMode(LED_PIN, OUTPUT); 
} 
  
void loop() 
{ 
  // выдаём неполное напряжение на светодиод 
  // (он же ШИМ-сигнал, он же PWM-сигнал). 
  // Микроконтроллер переводит число от 0 до 255 к напряжению 
  // от 0 до 5 В. Например, 85 — это 1/3 от 255, 
  // т.е. 1/3 от 5 В, т.е. 1,66 В. 
  analogWrite(LED_PIN, 85); 
  // держим такую яркость 250 миллисекунд 
  delay(250); 
  
  // выдаём 170, т.е. 2/3 от 255, или иными словами — 3,33 В. 
  // Больше напряжение — выше яркость! 
  analogWrite(LED_PIN, 170); 
  delay(250); 
  
  // все 5 В — полный накал! 
  analogWrite(LED_PIN, 255); 
  // ждём ещё немного перед тем, как начать всё заново 
  delay(250); 
} 

Задание на лабораторную работу 

Отключите питание, отключите светодиод от 9-го порта и подключите к 11-

му. Измените программу так, чтобы схема снова заработала 



Измените код программы так, чтобы в течение секунды на светодиод после-

довательно подавалось усреднённое напряжение 0, 1, 2, 3, 4, 5 В 

Возьмите еще один светодиод, резистор на 220 Ом и соберите аналогичную 

схему на этой же макетке, подключив светодиод к пину номер 3 и другому входу 

GND, измените программу так, чтобы светодиоды мигали в противофазу: первый 

выключен, второй горит максимально ярко и до противоположного состояния 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Какие из следующих идентификаторов корректны и не вызовут ошибку? 

13pin 

MOTOR_1 

контакт_светодиода 

sensor value 

leftServo 

my-var 

distance_eval2 

2. Что произойдет, если создать директиву #define HIGH LOW? 

3. Почему мы не сможем регулировать яркость светодиода, подключенного к порту 

7? 

4. Какое усреднённое напряжение мы получим на пине 6, если вызовем функцию 

analogWrite(6, 153)? 

5. Какое значение параметра value нужно передать функции analogWrite, чтобы по-

лучить усреднённое напряжение 2 В? 



Лабораторная работа №3 

Светильник с управляемой яркостью 

Цель работы: ознакомиться с принципами аналогового ввода. 

Краткие сведения из теории 

 
Рисунок 3.1 – Принципиальная схема 

 

Листинг программы: 
// даём разумные имена для пинов со светодиодом 
// и потенциометром (англ potentiometer или просто «pot») 
#define LED_PIN     9 
#define POT_PIN     A0 
  
void setup() 
{ 
  // пин со светодиодом — выход, как и раньше... 
  pinMode(LED_PIN, OUTPUT); 
  
  // ...а вот пин с потенциометром должен быть входом 
  // (англ. «input»): мы хотим считывать напряжение, 
  // выдаваемое им 
  pinMode(POT_PIN, INPUT); 
} 
  
void loop() 
{ 
  // заявляем, что далее мы будем использовать 2 переменные с 
  // именами rotation и brightness, и что хранить в них будем 
  // целые числа (англ. «integer», сокращённо просто «int») 
  int rotation, brightness;   
  
  // считываем в rotation напряжение с потенциометра: 
  // микроконтроллер выдаст число от 0 до 1023 
  // пропорциональное углу поворота ручки 
  rotation = analogRead(POT_PIN); 
  
  // в brightness записываем полученное ранее значение rotation 
  // делённое на 4. Поскольку в переменных мы пожелали хранить 
  // целые значения, дробная часть от деления будет отброшена. 
  // В итоге мы получим целое число от 0 до 255 
  brightness = rotation / 4; 
  
  // выдаём результат на светодиод 
  analogWrite(LED_PIN, brightness); 
} 

Задание на лабораторную работу 



Отключите питание платы, подключите к порту 5 еще один светодиод. Из-

мените код таким образом, чтобы второй светодиод светился на 1/8 от яркости пер-

вого 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Можем ли мы при сборке схемы подключить светодиод и потенциометр напря-

мую к разным входам GND микроконтроллера? 

2. В какую сторону нужно крутить переменный резистор для увеличения яркости 

светодиода? 

3. Что будет, если стереть из программы строчку pinMode(LED_PIN, OUTPUT)? 

строчку pinMode(POT_PIN, INPUT)? 

4. Зачем мы делим значение, полученное с аналогового входа перед тем, как задать 

яркость светодиода? что будет, если этого не сделать? 



Лабораторная работа №4 

Управление вводом/выводом 

Цель работы: ознакомиться с принципами ввода/вывода. 

Краткие сведения из теории 

 
Рисунок 4.1 – Принципиальная схема 

 

Листинг программы: 
// даём имена для пинов с пьезопищалкой (англ. buzzer) и фото- 
// резистором (англ. Light Dependent Resistor или просто LDR) 
#define BUZZER_PIN  3 
#define LDR_PIN     A0 
  
void setup() 
{ 
  // пин с пьезопищалкой — выход... 
  pinMode(BUZZER_PIN, OUTPUT); 
  
  // ...а все остальные пины являются входами изначально, 
  // всякий раз при подаче питания или сбросе микроконтроллера. 
  // Поэтому, на самом деле, нам совершенно необязательно 
  // настраивать LDR_PIN в режим входа: он и так им является 
} 
  
void loop() 
{ 
  int val, frequency; 
  
  // считываем уровень освещённости так же, как для 
  // потенциометра: в виде значения от 0 до 1023. 
  val = analogRead(LDR_PIN); 
  
  // рассчитываем частоту звучания пищалки в герцах (ноту), 
  // используя функцию проекции (англ. map). Она отображает 
  // значение из одного диапазона на другой, строя пропорцию. 
  // В нашем случае [0; 1023] -> [3500; 4500]. Так мы получим 
  // частоту от 3,5 до 4,5 кГц. 
  frequency = map(val, 0, 1023, 3500, 4500); 
  
  // заставляем пин с пищалкой «вибрировать», т.е. звучать 
  // (англ. tone) на заданной частоте 20 миллисекунд. При 
  // cледующих проходах loop, tone будет вызван снова и снова, 
  // и на деле мы услышим непрерывный звук тональностью, которая 
  // зависит от количества света, попадающего на фоторезистор 
  tone(BUZZER_PIN, frequency, 20); 
} 

Задание на лабораторную работу 



Уберите из программы чтение датчика освещенности и пропищите азбукой 

Морзе позывной SOS: три точки, три тире, три точки 

Измените код программы так, чтобы с падением освещенности звук стано-

вился ниже (например, падал от 5 кГц до 2,5 кГц) 

Измените код программы так, чтобы звук терменвокса раздавался не непре-

рывно, а 10 раз в секунду с различимыми паузами 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Каким сопротивлением должен обладать фоторезистор, чтобы на аналоговый 

вход было подано напряжение 1 В? 

2. Можем ли мы регулировать яркость светодиода, подключенного к 11-му порту, во 

время звучания пьезопищалки? 

3. Что изменится в работе терменвокса, если заменить резистор на 10 кОм резисто-

ром на 100 кОм? Попробуйте ответить без эксперимента. Затем отключите пита-

ние, замените резистор и проверьте. 

4. Каков будет результат вызова map(30,0,90,90,-90)? 

5. Как будет работать вызов tone без указания длительности звучания? 

6. Можно ли устроить полифоническое звучание с помощью функции tone? 



Лабораторная работа №5 

Управление выводом по событию на вводе 

Цель работы: ознакомиться с принципами ввода/вывода. 

Краткие сведения из теории 

 
Рисунок 5.1 – Принципиальная схема 

 

Листинг программы: 
#define LED_PIN  13 
#define LDR_PIN  A0 
#define POT_PIN  A1 
void setup() 
{ 
  pinMode(LED_PIN, OUTPUT); 
} 
  
void loop() 
{ 
  // считываем уровень освещённости. Кстати, объявлять 
  // переменную и присваивать ей значение можно разом 
  int lightness = analogRead(LDR_PIN); 
  // считываем значение с потенциометра, которым мы регулируем 
  // пороговое значение между условными темнотой и светом 
  int threshold = analogRead(POT_PIN); 
  // объявляем логическую переменную и назначаем ей значение 
  // «темно ли сейчас». Логические переменные, в отличие от 
  // целочисленных, могут содержать лишь одно из двух значений: 
  // истину (англ. true) или ложь (англ. false). Такие значения 
  // ещё называют булевыми (англ. boolean). 
  boolean tooDark = (lightness < threshold); 
  // используем ветвление программы: процессор исполнит один из 
  // двух блоков кода в зависимости от исполнения условия. 
  // Если (англ. «if») слишком темно... 
  if (tooDark) { 
    // ...включаем освещение 
    digitalWrite(LED_PIN, HIGH); 
  } else { 
    // ...иначе свет не нужен — выключаем его 
    digitalWrite(LED_PIN, LOW); 
  } 
} 



Задание на лабораторную работу 

Перепишите программу без использования переменной tooDark с сохранени-

ем функционала устройства. 

Добавьте в схему еще один светодиод. Дополните программу так, чтобы при 

падении освещенности ниже порогового значения включался один светодиод, а при 

падении освещенности ниже половины от порогового значения включались оба све-

тодиода. 

Измените схему и программу так, чтобы светодиоды включались по преж-

нему принципу, но светились тем сильнее, чем меньше света падает на фоторези-

стор. 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Если мы установим фоторезистор между аналоговым входом и землей, наше 

устройство будет работать наоборот: светодиод будет включаться при увеличении 

количества света. Почему? 

2. Какой результат работы устройства мы получим, если свет от светодиода будет 

падать на фоторезистор? 

3. Если мы все же установили фоторезистор так, как сказано в предыдущем вопросе, 

как нам нужно изменить программу, чтобы устройство работало верно? 

4. Допустим, у нас есть код if (условие) {действие;}. В каких случаях будет выпол-

нено действие? 

5. При каких значениях y выражение x + y > 0 будет истинным, если x > 0? 

6. Обязательно ли указывать, какие инструкции выполнять, если условие в операто-

ре if ложно? 

7. Чем отличается оператор == от оператора =? 

8. Если мы используем конструкцию if (условие) действие1; else действие2;, может 

ли быть ситуация, когда ни одно из действий не выполнится? Почему? 



Лабораторная работа №6 

Управление таймером с выводом на семисегментный индикатор 

Цель работы: ознакомиться с принципами вывода. 

Краткие сведения из теории 

 
Рисунок 6.1 – Принципиальная схема 

 

Листинг программы: 
#define FIRST_SEGMENT_PIN   2 
#define SEGMENT_COUNT       7 
// префикс «0b» означает, что целое число за ним записано в 
// в двоичном коде. Единицами мы обозначим номера сегментов 
// индикатора, которые должны быть включены для отображения 
// арабской цифры. Всего цифр 10, поэтому в массиве 10 чисел. 
// Нам достаточно всего байта (англ. byte, 8 бит) для хранения 
// комбинации сегментов для каждой из цифр. 
byte numberSegments[10] = { 
  0b00111111, 0b00001010, 0b01011101, 0b01011110, 0b01101010, 
  0b01110110, 0b01110111, 0b00011010, 0b01111111, 0b01111110, 
}; 
void setup() 
{ 
  for (int i = 0; i < SEGMENT_COUNT; ++i) 
    pinMode(i + FIRST_SEGMENT_PIN, OUTPUT); 
} 
void loop() 
{ 
  // определяем число, которое собираемся отображать. Пусть им 
  // будет номер текущей секунды, зацикленный на десятке 
  int number = (millis() / 1000) % 10; 
  // получаем код, в котором зашифрована арабская цифра 
  int mask = numberSegments[number]; 
  // для каждого из 7 сегментов индикатора... 
  for (int i = 0; i < SEGMENT_COUNT; ++i) { 
    // ...определяем: должен ли он быть включён. Для этого 
    // считываем бит (англ. read bit), соответствующий текущему 
    // сегменту «i». Истина — он установлен (1), ложь — нет (0) 
    boolean enableSegment = bitRead(mask, i); 
    // включаем/выключаем сегмент на основе полученного значения 
    digitalWrite(i + FIRST_SEGMENT_PIN, enableSegment); 
  } 
} 



Задание на лабораторную работу 

Измените код, чтобы индикатор отсчитывал десятые секунды. 

Поменяйте программу так, чтобы вместо символа «0» отображался символ 

«А». 

Дополните схему и программу таким образом, чтобы сегмент-точка вклю-

чался при прохождении четных чисел и выключался на нечетных 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. К которой ножке нашего семисегментного индикатора нужно подключать зем-

лю? 

2. Как мы храним закодированные символы цифр? 

3. Каким образом мы выводим символ на индикатор? 



Лабораторная работа №7 

Управление аналоговым выводом с помощью цифрового ввода 

Цель работы: ознакомиться с принципами ввода и вывода. 

Краткие сведения из теории 

 
Рисунок 7.1 – Принципиальная схема 

 

Листинг программы: 
#define PLUS_BUTTON_PIN     2 
#define MINUS_BUTTON_PIN    3 
#define LED_PIN             9 
int brightness = 100; 
boolean plusUp = true; 
boolean minusUp = true; 
void setup() 
{ 
  pinMode(LED_PIN, OUTPUT); 
  pinMode(PLUS_BUTTON_PIN, INPUT_PULLUP); 
  pinMode(MINUS_BUTTON_PIN, INPUT_PULLUP); 
} 
void loop() 
{ 
  analogWrite(LED_PIN, brightness); 
  // реагируем на нажатия с помощью функции, написанной нами 
  plusUp = handleClick(PLUS_BUTTON_PIN, plusUp, +35); 
  minusUp = handleClick(MINUS_BUTTON_PIN, minusUp, -35); 
} 
// Собственная функция с 3 параметрами: номером пина с кнопкой 
// (buttonPin), состоянием до проверки (wasUp) и градацией 
// яркости при клике на кнопку (delta). Функция возвращает 
// (англ. return) обратно новое, текущее состояние кнопки 
boolean handleClick(int buttonPin, boolean wasUp, int delta) 
{ 
  boolean isUp = digitalRead(buttonPin); 
  if (wasUp && !isUp) { 
    delay(10); 
    isUp = digitalRead(buttonPin); 
    // если был клик, меняем яркость в пределах от 0 до 255 
    if (!isUp)     
      brightness = constrain(brightness + delta, 0, 255); 
  } 
  return isUp; // возвращаем значение обратно, в вызывающий код 
} 

Задание на лабораторную работу 



Доработайте код таким образом, чтобы шаг изменения яркости настраивался 

в одном месте. 

Создайте еще одну функцию и переделайте код так, чтобы одна функция от-

вечала за отслеживание нажатий, а другая — за вычисление яркости светодиода и 

возвращала его в analogWrite. 

Содержание отчета 

В отчете указать цель работы, привести разработанные схему и листинг про-

граммы. 

Контрольные вопросы 

1. Что необходимо для определения собственной функции? 

2. Что означает ключевое слово void? 

3. Как ведет себя программа при упоминании одной переменной с разных сторон от 

оператора присваивания =? 

1 Подготовка к лабораторным занятиям 

Для того чтобы лабораторные занятия приносили максимальную пользу, 

необходимо помнить, что упражнение и решение задач проводятся по рассмотрен-

ному на лекциях материалу и связаны, как правило, с детальным разбором отдель-

ных вопросов лекционного курса. Следует подчеркнуть, что только после усвоения 

лекционного материала с определенной точки зрения (а именно с той, с которой он 

излагается на лекциях) он будет закрепляться студентом на лабораторных занятиях 

как в результате обсуждения и анализа лекционного материала, так и с помощью 

решения проблемных ситуаций, задач. При этих условиях студент не только хорошо 

усвоит материал, но и научится применять его на практике, а также получит допол-

нительный стимул (и это очень важно) для активной проработки лекции. 

При самостоятельном решении задач нужно обосновывать каждый этап реше-

ния, исходя из теоретических положений курса. Если студент видит несколько пу-

тей решения проблемы (задачи), то нужно сравнить их и выбрать самый рациональ-

ный. Полезно до начала вычислений составить краткий план решения проблемы (за-

дачи). Решение проблемных задач или примеров следует излагать подробно, вычис-

ления располагать в строгом порядке, отделяя вспомогательные вычисления от ос-

новных. Решения при необходимости нужно сопровождать комментариями, схема-

ми, чертежами и рисунками. 

Следует помнить, что решение каждой учебной задачи должно доводиться до 

окончательного логического ответа, которого требует условие, и по возможности с 

выводом. Полученный ответ следует проверить способами, вытекающими из суще-

ства данной задачи. Полезно также (если возможно) решать несколькими способами 

и сравнить полученные результаты. Решение задач данного типа нужно продолжать 

до приобретения твердых навыков в их решении. 

2 Подготовка к практическим занятиям 

Подготовку к каждому практическому занятию студент должен начать с озна-

комления с методическими указаниями, которые включают содержание работы. 

Тщательное продумывание и изучение вопросов основывается на проработке теку-

щего материала лекции, а затем изучения обязательной и дополнительной литерату-

ры, рекомендованную к данной теме. На основе индивидуальных предпочтений сту-

денту необходимо самостоятельно выбрать тему доклада по проблеме и по возмож-

ности подготовить по нему презентацию.  

Если программой дисциплины предусмотрено выполнение практического за-

дания, то его необходимо выполнить с учетом предложенной инструкции (устно или 

письменно). Все новые понятия по изучаемой теме необходимо выучить наизусть и 

внести в глоссарий, который целесообразно вести с самого начала изучения курса. 

Результат такой работы должен проявиться в способности студента свободно отве-

тить на теоретические вопросы семинара, его выступлении и участии в коллектив-

ном обсуждении вопросов изучаемой темы, правильном выполнении практических 

заданий и контрольных работ.  

В зависимости от содержания и количества отведенного времени на изучение 

каждой темы практическое занятие может состоять из четырех-пяти частей:  

1. Обсуждение теоретических вопросов, определенных программой дисци-

плины.  

2. Доклад и/ или выступление с презентациями по выбранной проблеме.  

3. Обсуждение выступлений по теме – дискуссия.  

4. Выполнение практического задания с последующим разбором полученных 

результатов или обсуждение практического задания. 

5. Подведение итогов занятия.  

Первая часть – обсуждение теоретических вопросов – проводится в виде 

фронтальной беседы со всей группой и включает выборочную проверку преподава-

телем теоретических знаний студентов. Примерная продолжительность — до 15 ми-

нут. Вторая часть — выступление студентов с докладами, которые должны сопро-

вождаться презентациями с целью усиления наглядности восприятия, по одному из 

вопросов практического занятия. Обязательный элемент доклада – представление и 

анализ статистических данных, обоснование социальных последствий любого эко-

номического факта, явления или процесса. Примерная продолжительность — 20-25 

минут. После докладов следует их обсуждение – дискуссия. В ходе этого этапа 

практического занятия могут быть заданы уточняющие вопросы к докладчикам. 

Примерная продолжительность – до 15-20 минут. Если программой предусмотрено 

выполнение практического задания в рамках конкретной темы, то преподавателями 

определяется его содержание и дается время на его выполнение, а замет идет об-

суждение результатов. Подведением итогов заканчивается практическое занятие.  

В процессе подготовки к практическим занятиям, студентам необходимо об-

ратить особое внимание на самостоятельное изучение рекомендованной учебно-

методической (а также научной и популярной) литературы. Самостоятельная работа 

с учебниками, учебными пособиями, научной, справочной и популярной литерату-

рой, материалами периодических изданий и Интернета, статистическими данными 

является наиболее эффективным методом получения знаний, позволяет значительно 

активизировать процесс овладения информацией, способствует более глубокому 

усвоению изучаемого материала, формирует у студентов свое отношение к конкрет-

ной проблеме. Более глубокому раскрытию вопросов способствует знакомство с до-

полнительной литературой, рекомендованной преподавателем по каждой теме се-

минарского или практического занятия, что позволяет студентам проявить свою ин-

дивидуальность в рамках выступления на данных занятиях, выявить широкий 

спектр мнений по изучаемой проблеме. 

3 Самостоятельное изучение темы. Конспект 

Конспект – наиболее совершенная и наиболее сложная форма записи. Слово 

«конспект» происходит от латинского «conspectus», что означает «обзор, изложе-

ние». В правильно составленном конспекте обычно выделено самое основное в изу-

чаемом тексте, сосредоточено внимание на наиболее существенном, в кратких и 

четких формулировках обобщены важные теоретические положения.  

Конспект представляет собой относительно подробное, последовательное из-

ложение содержания прочитанного. На первых порах целесообразно в записях бли-

же держаться тексту, прибегая зачастую к прямому цитированию автора. В даль-

нейшем, по мере выработки навыков конспектирования, записи будут носить более 

свободный и сжатый характер.  

Конспект книги обычно ведется в тетради. В самом начале конспекта указыва-

ется фамилия автора, полное название произведения, издательство, год и место из-

дания. При цитировании обязательная ссылка на страницу  книги. Если цитата взята 

из собрания сочинений, то необходимо указать соответствующий том. Следует пом-

нить, что четкая ссылка на источник – непременное правило конспектирования. Ес-

ли конспектируется статья, то указывается, где и когда она была напечатана. 

Конспект подразделяется на части в соответсвтии с заранее продуманным 

планом. Пункты плана записываются в тексте или на полях конспекта. Писать его 

рекомендуется четко и разборчиво, так как небрежная запись с течением времени 

становиться малопонятной для ее автора. Существует правило: конспект, составлен-

ный для себя, должен быть по возможности написан так, чтобы его легко прочитал и 

кто-либо другой. 

Формы конспекта могут быть разными и зависят от его целевого назначения 

(изучение материала в целом или под определенным углом зрения, подготовка к до-

кладу, выступлению на занятии и т.д.), а также от характера произведения (моно-

графия, статья, документ и т.п.). Если речь идет просто об изложении содержания 

работы, текст конспекта может быть сплошным, с выделением особо важных поло-

жений подчеркиванием или различными значками. 

В случае, когда  не ограничиваются переложением содержания, а фиксируют в 

конспекте и свои собственные суждения по данному вопросу или дополняют кон-

спект соответствующими материалами их других источников, следует отводить ме-

сто для такого рода записей. Рекомендуется разделить страницы тетради пополам по 

вертикали и в левой части вести конспект произведения, а в правой свои дополни-

тельные записи, совмещая их по содержанию. 

Конспектирование в большей мере, чем другие виды записей, помогает выра-

батывать навыки правильного изложения в письменной форме важные теоретиче-

ских и практических вопросов, умение четко их формулировать и ясно излагать сво-

ими словами. 

Таким образом, составление конспекта требует вдумчивой работы, затраты 

времени и труда. Зато во время конспектирования приобретаются знания, создается 

фонд записей. 

Конспект может быть текстуальным или тематическим. В текстуальном кон-

спекте сохраняется логика и структура изучаемого произведения, а запись ведется в 

соответствии с расположением материала в книге.  За основу тематического кон-

спекта берется не план произведения, а содержание какой-либо темы или проблемы. 

Текстуальный конспект желательно начинать после того, как вся книга прочи-

тана и продумана, но это, к сожалению, не всегда возможно. В первую очередь 

необходимо составить план произведения письменно или мысленно, поскольку в 

соответствии с этим планом строится дальнейшая работа. Конспект включает в себя 

тезисы, которые составляют его основу. Но, в отличие от тезисов, конспект содер-

жит краткую запись не только выводов, но и доказательств, вплоть до фактического 

материала. Иначе говоря, конспект – это расширенные тезисы, дополненные рас-

суждениями и доказательствами, мыслями и соображениями составителя  записи. 

Как правило, конспект включает в себя и  выписки, но в него могут войти от-

дельные места, цитируемые дословно, а также факты, примеры, цифры, таблицы и 

схемы, взятые из книги. Следует помнить, что работа над конспектом только тогда 

будет творческой, когда она не ограничена текстом изучаемого произведения. Нуж-

но дополнять конспект данными из другими источников. 

В конспекте необходимо выделять отдельные места текста в зависимости от 

их значимости. Можно пользоваться различными способами: подчеркиваниями, во-

просительными и восклицательными знаками, репликами, краткими оценками, пи-

сать на полях своих конспектов слова: «важно», «очень важно», «верно», «характер-

но». 

В конспект могут помещаться диаграммы, схемы, таблицы, которые придадут 

ему наглядность. 

Составлению тематического конспекта предшествует тщательное изучение 

всей литературы, подобранной для раскрытия данной темы. Бывает, что какая-либо 

тема рассматривается в нескольких главах или в разных местах книги. А в конспекте 

весь материал, относящийся к теме, будет сосредоточен в одном месте. В плане кон-

спекта рекомендуется делать пометки, к каким источникам (вплоть до страницы) 

придется обратиться для раскрытия вопросов. Тематический конспект составляется 

обычно для того, чтобы глубже изучить определенный вопрос, подготовиться к до-

кладу, лекции или выступлению на семинарском занятии. Такой конспект по содер-

жанию приближается к реферату, докладу по избранной теме, особенно если вклю-

чает и собственный вклад в изучение проблемы. 

4 Тематическое задание для выполнения контрольной работы 

 
Задача 

Два числа: номер студенческого билета число А и год рождения число Б. На 

Arduino не используя встроенных функций и библиотек написать программный мо-

дуль который переведет эти числа в двоичную систему счисления и результат выве-

дет с помощью светодиодов.  

Оборудование: 

− Arduino 

− макетная плата 

− два светодиода 

− два резистора номиналом 220 Ом 

− соединительные провода 

− USB-кабель 

